NAG Library Routine Document

G05RCF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05RCF sets up a reference vector and generates an array of pseudorandom numbers from a Student's t copula with ν degrees of freedom and covariance matrix $\frac{\nu}{\nu-2}C$.

2 Specification

```
SUBROUTINE G05RCF (MODE, N, DF, M, C, LDC, R, LR, STATE, X, LDX, IFAIL)

INTEGER             MODE, N, DF, M, LDC, LR, STATE(*), LDX, IFAIL
REAL (KIND=nag_wp) C(LDC,M), R(LR), X(LDX,M)
```

3 Description

The Student's t copula, G, is defined by

$$G(u_1, u_2, \dots, u_m; C) = T^m_{\nu,C}\left(t^{-1}_{\nu,C_{11}}(u_1), t^{-1}_{\nu,C_{22}}(u_2), \dots, t^{-1}_{\nu,C_{mm}}(u_m)\right)$$

where m is the number of dimensions, $T^m_{\nu,C}$ is the multivariate Student's t density function with ν degrees of freedom, mean zero and covariance matrix $\frac{\nu}{\nu-2}C$ and $t^{-1}_{\nu,C_{ii}}$ is the inverse of the univariate Student's t density function with ν degrees of freedom, zero mean and variance $\frac{\nu}{\nu-2}C_{ii}$.

G05RYF is used to generate a vector from a multivariate Student's t distribution and G01EBF is used to convert each element of that vector into a uniformly distributed value between zero and one.

One of the initialization routines G05KFF (for a repeatable sequence if computed sequentially) or G05KGF (for a non-repeatable sequence) must be called prior to the first call to G05RCF.

4 References

Nelsen R B (1998) *An Introduction to Copulas. Lecture Notes in Statistics 139* Springer

Sklar A (1973) Random variables: joint distribution functions and copulas *Kybernetika* **9** 499–460

5 Parameters

1: MODE – INTEGER *Input*

On entry: a code for selecting the operation to be performed by the routine.

MODE $= 0$
  Set up reference vector only.

MODE $= 1$
  Generate variates using reference vector set up in a prior call to G05RCF.

MODE $= 2$
  Set up reference vector and generate variates.

Constraint: MODE $= 0$, 1 or 2.

2: N – INTEGER *Input*

On entry: n, the number of random variates required.

Constraint: $N \geq 0$.

3: DF – INTEGER *Input*

On entry: ν, the number of degrees of freedom of the distribution.

Constraint: $DF \geq 3$.

4: M – INTEGER *Input*

On entry: m, the number of dimensions of the distribution.

Constraint: $M > 0$.

5: C(LDC,M) – REAL (KIND=nag_wp) array *Input*

On entry: matrix which, along with DF, defines the covariance of the distribution. Only the upper triangle need be set.

Constraint: C must be positive semidefinite to ***machine precision***.

6: LDC – INTEGER *Input*

On entry: the first dimension of the array C as declared in the (sub)program from which G05RCF is called.

Constraint: $LDC \geq M$.

7: R(LR) – REAL (KIND=nag_wp) array *Communication Array*

On entry: if $MODE = 1$, the reference vector as set up by G05RCF in a previous call with $MODE = 0$ or 2.

On exit: if $MODE = 0$ or 2, the reference vector that can be used in subsequent calls to G05RCF with $MODE = 1$.

8: LR – INTEGER *Input*

On entry: the dimension of the array R as declared in the (sub)program from which G05RCF is called. If $MODE = 1$, it must be the same as the value of LR specified in the prior call to G05RCF with $MODE = 0$ or 2.

Constraint: $LR \geq M \times (M + 1) + 2$.

9: STATE($*$) – INTEGER array *Communication Array*

Note: the actual argument supplied must be the array STATE supplied to the initialization routines G05KFF or G05KGF.

On entry: contains information on the selected base generator and its current state.

On exit: contains updated information on the state of the generator.

10: X(LDX,M) – REAL (KIND=nag_wp) array *Output*

On exit: the array of values from a multivariate Student's t copula, with $X(i,j)$ holding the jth dimension for the ith variate.

11: LDX – INTEGER *Input*

On entry: the first dimension of the array X as declared in the (sub)program from which G05RCF is called.

Constraint: $LDX \geq N$.

12: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, MODE $\neq 0$, 1 or 2.

IFAIL $= 2$

On entry, N < 0.

IFAIL $= 3$

On entry, DF ≤ 2.

IFAIL $= 4$

On entry, M < 1.

IFAIL $= 5$

The covariance matrix C is not positive semidefinite to ***machine precision***.

IFAIL $= 6$

On entry, LDC $<$ M.

IFAIL $= 7$

The reference vector R has been corrupted or M has changed since R was set up in a previous call to G05RCF with MODE $= 0$ or 2.

IFAIL $= 8$

On entry, LR $\leq \mathrm{M} \times (\mathrm{M}+1)+1$.

IFAIL $= 9$

On entry, STATE vector was not initialized or has been corrupted.

IFAIL $= 11$

On entry, LDX $<$ N.

7 Accuracy

See Section 7 in G05RYF for an indication of the accuracy of the underlying multivariate Student's t-distribution.

8 Further Comments

The time taken by G05RCF is of order nm^3.

It is recommended that the diagonal elements of C should not differ too widely in order of magnitude. This may be achieved by scaling the variables if necessary. The actual matrix decomposed is $C + E = LL^{\mathrm{T}}$, where E is a diagonal matrix with small positive diagonal elements. This ensures that, even when C is singular, or nearly singular, the Cholesky factor L corresponds to a positive definite covariance matrix that agrees with C within ***machine precision***.

9 Example

This example prints ten pseudorandom observations from a Student's t copula with ten degrees of freedom and C matrix

$$\begin{bmatrix} 1.69 & 0.39 & -1.86 & 0.07 \\ 0.39 & 98.01 & -7.07 & -0.71 \\ -1.86 & -7.07 & 11.56 & 0.03 \\ 0.07 & -0.71 & 0.03 & 0.01 \end{bmatrix},$$

generated by G05RCF. All ten observations are generated by a single call to G05RCF with MODE $= 2$. The random number generator is initialized by G05KFF.

9.1 Program Text

```
    Program g05rcfe

!     G05RCF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: g05kff, g05rcf, nag_wp, x04caf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: lseed = 1, nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                          :: df, genid, i, ifail, ldc, ldx, lr,   &
                                          lstate, m, mode, n, subid
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: c(:,:), r(:), x(:,:)
      Integer                          :: seed(lseed)
      Integer, Allocatable             :: state(:)
!     .. Executable Statements ..
      Write (nout,*) 'G05RCF Example Program Results'
      Write (nout,*)
      Flush (nout)

!     Skip heading in data file
      Read (nin,*)

!     Read in the base generator information and seed
      Read (nin,*) genid, subid, seed(1)

!     Initial call to initialiser to get size of STATE array
      lstate = 0
      Allocate (state(lstate))
      ifail = 0
      Call g05kff(genid,subid,seed,lseed,state,lstate,ifail)
```

```
!     Reallocate STATE
      Deallocate (state)
      Allocate (state(lstate))

!     Initialize the generator to a repeatable sequence
      ifail = 0
      Call g05kff(genid,subid,seed,lseed,state,lstate,ifail)

!     Read in sample size and number of dimensions
      Read (nin,*) n, m

      ldc = m
      ldx = n
      lr = m*(m+1) + 2
      Allocate (c(ldc,m),x(ldx,m),r(lr))

!     Read in degrees of freedom
      Read (nin,*) df

!     Read in upper triangle portion of the covariance matrix
      Do i = 1, m
        Read (nin,*) c(i,i:m)
      End Do

!     Using a single call to G05RCF, so set up reference vector
!     and generate values in one go
      mode = 2

!     Generate variates
      ifail = 0
      Call g05rcf(mode,n,df,m,c,ldc,r,lr,state,x,ldx,ifail)

!     Display the variates
      ifail = 0
      Call x04caf('General',' ',n,m,x,ldx,'Variates',ifail)

    End Program g05rcfe
```

9.2 Program Data

```
G05RCF Example Program Data
1  1  1762543          :: GENID,SUBID,SEED(1)
10  4                  :: N,M
10                     :: DF
1.69  0.39 -1.86  0.07
     98.01 -7.07 -0.71
           11.56  0.03
                  0.01 :: End of C (upper triangular part)
```

9.3 Program Results

```
 G05RCF Example Program Results

 Variates
           1       2       3       4
  1   0.6445  0.0527  0.4082  0.8876
  2   0.0701  0.1988  0.8471  0.3521
  3   0.7988  0.6664  0.2194  0.5541
  4   0.8202  0.0492  0.7059  0.9341
  5   0.1786  0.5594  0.7810  0.2836
  6   0.4920  0.2677  0.3427  0.5169
  7   0.4139  0.2978  0.8762  0.7145
  8   0.7437  0.9714  0.8931  0.2487
  9   0.4971  0.9687  0.8142  0.1965
 10   0.6464  0.5304  0.5817  0.4565
```